§ 12.4.1: Anpassung einer GF an neue RB.

Berechnung der GFw der zeitunabhängigen Diffusionsgleichung in [0,a] aus Go, der fuer [-∞,∞].
k = κ.

■ Greensche Funktionen

```
go = 1 / 2 / k Exp[- k Abs[x - xp]]
```

$$\frac{e^{-k\,\mathrm{Abs}[x-xp]}}{2\,k}$$

```
gn = go + c1 Exp[k x] + c2 Exp[-k x]
```

$$c2\,e^{-k\,x}+c1\,e^{k\,x}+\frac{e^{-k\,\mathrm{Abs}[x-xp]}}{2\,k}$$

■ RB bei x = 0:

```
eq1 = Simplify[gn, x < xp] == 0 /. x → 0
```

$$c1+c2+\frac{e^{-k\,xp}}{2\,k}==0$$

■ RB bei x = a:

```
eq2 = Simplify[gn, x > xp] == 0 /. x → a
```

$$c2\,e^{-a\,k}+c1\,e^{a\,k}+\frac{e^{k\,(-a+xp)}}{2\,k}==0$$

■ Berechnung der Koeffizienten

```
so = Solve[{eq1, eq2}, {c1, c2}] // Flatten
```

$$\left\{c1\to-\frac{e^{-k\,xp}\,\left(-1+e^{a\,k+k\,xp+k\,(-a+xp)}\right)}{2\,\left(-1+e^{2\,a\,k}\right)\,k},\;c2\to-\frac{e^{a\,k-k\,xp}\,\left(e^{a\,k}-e^{k\,xp+k\,(-a+xp)}\right)}{2\,\left(-1+e^{2\,a\,k}\right)\,k}\right\}$$

■ Berechnung von G2 für x > x'

```
so1 = ExpToTrig[Simplify[gn, x > xp] /. so ] // FullSimplify
```

$$\frac{\mathrm{Csch}[a\,k]\,\mathrm{Sinh}[k\,(a-x)]\,\mathrm{Sinh}[k\,xp]}{k}$$

■ Berechnung von G2 für x < x'

```
so2 = ExpToTrig[Simplify[gn, x < xp] /. so ] // FullSimplify
```

$$\frac{\mathrm{Csch}[a\,k]\,\mathrm{Sinh}[k\,x]\,\mathrm{Sinh}[k\,(a-xp)]}{k}$$

Csch[a k] = 1/Sinh[a k]. Man sieht aus den beiden letzten Resultaten:

$$G(x,x') = \frac{\sinh[k\,x_<]\,\sinh[k\,\{a-x_>)]}{k\,\sinh[a\,k]}$$